


Orbital motion

This exercise explores the motion of a particle in a spherical Coulomb potential. The equations of motion are similar to those for a planet orbiting the sun: the radial acceleration varies inversely with the square of the distance. Or, we could be simulating a classical hydrogen atom.

```
ORIGIN := 1
XXXXXXXXXX
```

This specifies that the vector components are numbered 1, 2 and 3 and not 0, 1, and 2.



Pick starting X and V vectors and a starting time.

$$X := \begin{pmatrix} 0 \\ 1 \\ 0 \end{pmatrix} \quad V := \begin{pmatrix} 0.75 \\ 0 \\ 0 \end{pmatrix} \quad t := 0$$

We start at $y = 1$ on the y axis where x and z are zero. The starting velocity is in the x direction so that we will orbit in the x,y plane.

We will stack the X and V values and put them into our 6-vector z $Z := \text{stack}(X, V)$

$$Z = \begin{pmatrix} 0 \\ 1 \\ 0 \\ 0.75 \\ 0 \\ 0 \end{pmatrix} \begin{matrix} x \\ y \\ z \\ Vx \\ Vy \\ Vz \end{matrix}$$

The next two submatrix definitions let us recover X and V from the stack Z for plotting:

$$X(Z) := \text{submatrix}(Z, 1, 3, 1, 1) \quad V(Z) := \text{submatrix}(Z, 4, 6, 1, 1)$$

We do this because it is easier to remember X and V than to remember the components of Z. We must have a 6-vector Z because the Runge-Kutta routine must have everything it needs in one variable.

The force will vary as $1/r^2$. The square of the radius is simply the dot product of X with itself. The force in the x direction is the radial force multiplied by x/r .

The radial acceleration will be C/r^2 with C selected as: $C := -1$

The x acceleration, for example, is $a_x := \frac{C}{r^2} \cdot \frac{x}{r}$ which is the same as $\frac{C \cdot Z_1}{(\sqrt{X \cdot X})^3}$

DZ holds the derivative of the vector Z.
 The top three terms are the velocities.
 The bottom three terms are the radial acceleration projected onto the x, y and z directions.

$$DZ(t,Z) := \begin{bmatrix} Z_4 \\ Z_5 \\ Z_6 \\ \frac{C \cdot Z_1}{(X(Z) \cdot X(Z))^{1.5}} \\ \frac{C \cdot Z_2}{(X(Z) \cdot X(Z))^{1.5}} \\ \frac{C \cdot Z_3}{(X(Z) \cdot X(Z))^{1.5}} \end{bmatrix} \begin{matrix} dx/dt \\ dy/dt \\ dz/dt \\ dV_x/dt \\ dV_y/dt \\ dV_z/dt \end{matrix}$$

Now let's find and plot the particle trajectory

Our time interval should be divided more finely than the gyration period which is 2π if we start with $v = 1$ at $r = 1$.

Let t be the total time interval $t := 5$

`npoints := ceil(20·t)` It is always a good idea to base npoints on a calculation using t. Then you can change t and the points will still be the same distance apart and you will have the same number of points per orbit.


Now integrate: $F := \text{rkfixed}(Z, 0, t, npoints, DZ)$

This is the trajectory in our 6-dimensional space:

| | t | x | y | z | V_x | V_y | V_z |
|-----|----|------|-------|-------|-------|-------|--------|
| | 1 | 2 | 3 | 4 | 5 | 6 | 7 |
| F = | 1 | 0 | 0 | 1 | 0 | 0.75 | 0 |
| | 2 | 0.05 | 0.037 | 0.999 | 0 | 0.749 | -0.05 |
| | 3 | 0.1 | 0.075 | 0.995 | 0 | 0.746 | -0.1 |
| | 4 | 0.15 | 0.112 | 0.989 | 0 | 0.742 | -0.15 |
| | 5 | 0.2 | 0.149 | 0.98 | 0 | 0.735 | -0.2 |
| | 6 | 0.25 | 0.186 | 0.969 | 0 | 0.726 | -0.251 |
| | 7 | 0.3 | 0.222 | 0.955 | 0 | 0.715 | -0.301 |
| | 8 | 0.35 | 0.257 | 0.939 | 0 | 0.703 | -0.352 |
| | 9 | 0.4 | 0.292 | 0.92 | 0 | 0.688 | -0.403 |
| | 10 | 0.45 | 0.326 | 0.898 | 0 | 0.67 | -0.455 |

Click on the table to show the scroll bar.


Our orbit reminds us of Kepler's observation that planetary orbits are ellipses with the sun at one focus.



F⁽²⁾ is x and F⁽³⁾ is y.

A zero is plotted on both axes so we can find the origin easily.

An interactive 3-d view, just grab a corner and pull:



$(F^{(2)}, F^{(3)}, F^{(4)})$

What happens if our orbiter is subjected to a small additional acceleration not in the plane of the orbit?

We will create a small perturbation with a force of 0.04 in the y direction and 0.04 in the z direction. Remember that we orbit in the x-y plane.


`t := 50` `npoints := ceil(20·t)` `gy := 0.04` `gz := 0.04`

$$DZ(t,Z) := \begin{bmatrix} Z_4 \\ Z_5 \\ Z_6 \\ \frac{C \cdot Z_1}{(X(Z) \cdot X(Z))^{1.5}} \\ \frac{C \cdot Z_2}{(X(Z) \cdot X(Z))^{1.5}} + gy \\ \frac{C \cdot Z_3}{(X(Z) \cdot X(Z))^{1.5}} + gz \end{bmatrix}$$

Here we add the small accelerations to the y and z directions.

Calculate and plot the new orbit in color (shaded in the z direction) and the original orbit in black:


`F2 := rkfixed(Z,0,t,npoints,DZ)`



$(F2^{(2)}, F2^{(3)}, F2^{(4)}), (F^{(2)}, F^{(3)}, F^{(4)})$

The plot is "too busy" to understand. Let's look at the original unperturbed orbit and the last few perturbed orbits. The last orbits we find from the last 200 points of F2 which we put in a submatrix F3 for plotting.

```
F3 := submatrix(F2, npoints - 200, npoints, 1, 7)
```



$$(F3^{(2)}, F3^{(3)}, F3^{(4)}), (F^{(2)}, F^{(3)}, F^{(4)})$$

Now it is clearer what is happening. The perturbation causes the orbit to slowly tilt over. The last orbit that we see is nearly perpendicular to the original orbit.

Try it: What happens if the potential varies with a different power of radius? (You can use "Replace" from the edit menu to change 1.5 everywhere to 1.2.)

Try it: What happens if the initial velocity is changed from 0.75 to 1.0?

The long axis of an elliptical orbit can slowly change its orientation. What would cause that? Or, an orbit can wobble (have some oscillator motion along z). What would cause that? Which is precession (tricky!)?

[These are not an easy questions. You can make the initial orbit tilted by giving a little z velocity, 0.2 for example, at the beginning.]